

# Computer-Simulation für Einsteiger

Die Zauberformel, die Sie dazu befähigt, steht in Zeile 200. Da auch ein gelernter Ingenieur oder Mathematiklehrer darin nicht auf Anhieb die Schwingungs-Differentialgleichung wiedererkennen kann, wird für mathematisch interessierte Leser im Texteingang die Umwandlung erklärt.

Das eigentliche Simulationsprogramm besteht aus nur fünf Zeilen (Listing 1). Alle eingerückten Programmzeilen können Sie zunächst vergessen, sie dienen nur dazu, die Lösungen auf dem Bildschirm darzustellen. Die vier Konstanten des Schwingungssystems sind in Zeile 130 zusammengefaßt. Ihre physikalische Bedeutung soll anhand der Abbildung 1 erklärt werden:

Eine Masse  $m$  schwingt auf einer Feder (Federstärke  $k$ ) in  $X$ -Richtung auf und ab. Die Frequenz dieser Schwingung ist von dem Verhältnis  $m/k$  abhängig. Wir nennen es, etwas unkonventionell, EFQ (Eigenfrequenz zum Quadrat).

Durch unvermeidliche oder absichtliche Reibung wird dem System Energie entzogen. Die Konstante  $D$  ist ein Maß für diese Schwingungsdämpfung. Ferner haben wir noch vorgesehen, daß das System von außen mit einer Stärke  $A$  und einer Frequenz  $WA$  angeregt werden kann. Es führt dann erzwungene Schwingungen aus ( $A$  hat die Dimension Kraft/Masse).

Um aus der Vielfalt der möglichen Schwingungen, die ein solches System ausführen kann, eine bestimmte auszusuchen, müssen wir noch in Zeile 140 die Anfangsbedingungen vorgeben. Sie legen fest, welche Auslenkung  $X$  und welche Geschwindigkeit  $V$  das Sy-

**Wer den interessanten Simulations-Artikel im Januarheft gelesen hat, könnte glauben, Computer-Simulation sei nur etwas für Experten. Aber auch auf dem Commodore 64 ist dies möglich.**

stem zur Zeit  $t=0$  haben soll. Der Computer berechnet aus den Anfangsbedingungen  $X(0)$  und  $V(0)$ , näherungsweise die Größen  $X(dt)$  und  $V(dt)$  zu einem etwas späteren Zeitpunkt  $dt$  und daraus wiederum die Werte  $X(2*dt)$  und  $V(2*dt)$  zwei Zeit-

einheiten später. So hangelt sich der Rechner von einem Zeitabschnitt zum nächsten. Diese Näherungsmethode ist schon von Euler vor 200 Jahren vorgeschlagen worden, aber sie ist so rechenintensiv, daß sie erst im Computer-Zeitalter bequem ausgeführt werden kann.

Der Rest des Basic-Programms dient, wie gesagt, nur dazu, die berechneten  $X$ -Werte in Abhängigkeit von der Zeit auf dem Bildschirm darzustellen. Die Zeit kann man durch die Zahl  $N$  der Zeiteinheiten ausdrücken:  $t=N*dt$ .

$dt$  haben wir willkürlich gleich eins gesetzt. Durch diesen Kunstgriff können wir auf der Zeitachse einfach unsere Schleifenvariable  $N$  (Zeile 190) auftragen und  $N$  von 0 bis 319 laufen lassen.

Es werden nur drei Grafikbefehle verwendet: Grafik einschalten, Punkt und Linie zeichnen. Sie können sie leicht an Ihr Grafiksystem anpassen.

Damit die  $X$ -Werte im Rahmen der 200 Bildschirmpunkte bleiben, muß der

Maßstabfaktor  $XM$  passend gewählt werden. Wenn nicht anders angegeben, arbeiten wir mit  $XM=1$ .

Zunächst müssen wir die Zuverlässigkeit unserer Zauberformel auf die Probe stellen. Schließlich handelt es sich um eine Näherungsrechnung, in der sich eventuelle Fehler über 320 Stufen summieren können. Wir simulieren zu diesem Zweck eine ungedämpfte Schwingung, von der wir wissen, daß alle Maxima gleich hoch sind und die Nulldurchgänge in gleichen Abständen auftreten müssen.

Ersetzen Sie bitte die Zeilen. 130 EFQ=.01:D=0:A=0:WA=0

140 V=0:X=80

Bild 2 zeigt, daß das Verfahren das richtige Ergebnis liefert.


Nach so viel Theorie wollen wir unsere Zauberformel nun auf ein praktisches Problem anwenden: Wie stark müssen die Stoßdämpfer eines Autos gedämpft sein, damit sie Stöße von Fahrbahnunebenheiten abfangen, aber nicht mehrmals auf und abspringen? (Bild 3). Das würde nämlich die Bodenhaftung und damit die Lenk- und Bremsfähigkeit verschlechtern, und außerdem den Fahrgästen auf den Magen schlagen.

130 EFQ=.01:D=(Werte siehe Text):A=0:WA=0


Für die Dämpfung  $D$  suchen Sie durch Probieren einen Wert, bei dem die Kurve die Zeitachse nicht kreuzt. Am besten versuchen Sie es mal mit Werten zwischen 0,01 und 0,5.

Vorher müssen Sie noch die Anfangsbedingungen in Zeile 140 eintragen. Wir nehmen an, daß die Feder um die Strecke -80 zusammengedrückt und zur Zeit  $t=0$  losgelassen wird.

140 V=0:X=-80



**Bild 1. Modell des Schwingungssystems  $m$ =Masse,  $k$ =Federkonstante,  $D$ =Dämpfung,  $A$ =Anregung,  $WA$ =Frequenz**



**Bild 2. Kurve der ungedämpften Schwingung**




Bild 3. Radaufhängung beim PKW (schematisch)




Bild 5. Aufbau einer Waage (schematisch)

Ihre Kurven werden einen Verlauf wie in Bild 4 haben, und Sie können feststellen, daß für  $D > 0,2$  kein Überschwingen stattfindet.

So, nun klettern Sie mal auf die Stoßstange Ihres Autos. Schwingt der Wagen nach dem Abspringen auf und ab?

Die richtige Dämpfung spielt auch beim zweiten Simulationsbeispiel eine Rolle: Der Zeiger einer Waage oder eines Meßinstrumentes soll möglichst schnell und ohne Hinunderschwingen den richtigen Wert anzeigen. Bild 5, eine Waage. Zur Zeit  $t = 0$  legen Sie einen Gegenstand auf die Waage und üben eine konstante Anregung A auf die Feder aus. In Zeile 200 müssen Sie das Sinusglied löschen und durch ein einfaches A ersetzen.

```
200 V=V-D*V-EFQ*X-A*A: X=X+V
130 EFQ=.01:D=(Werte .1, .2, .5):A=.5:WA=0
140 V=0:X=0
```

Das Ergebnis sehen Sie in Bild 6. Ist die Dämpfung zu klein (Kurve A), dann schwingt das System einige Zeit um die neue Ruhelage, ist sie zu groß (B), dann kriecht der Zeiger langsam dem Endausschlag entgegen. Hier kann es zu Fehlmessungen kommen, wenn man nicht lange genug wartet. Dazwischen gibt es eine optimale Dämpfung (C), bei der der Zeiger auf dem kürzesten Wege den Endwert ansteuert. Man kann die Grenzdämpfung aus der Eigenfrequenz berechnen:  $D = 2 * \text{SQR}(EFQ) = 0.2$ .

Beim dritten Simulationsbeispiel setzen wir einen Motor auf ein Masse-Feder-

```
100 : REM * SIMULATION VON SCHWINGUNGSVORGA
      ENGEN *
113 : REM * FUER C16,C116 UND C128
120 : REM * KENNGROESSEN DES SCHWINGUNGSSYS
      TEMS *
130 EFQ=.01: D=0: A=0: WA=0
140 V=0: X=0: REM * ANFANGSBEDINGUNGEN *
150 : XM=1: C=0: REM * X-MASSSTAB *
160 GRAPHIC 1,1
170 DRAW 1,0,100 TO 319,100
180 DRAW 1,0,0 TO 0,199
190 FOR N=1 TO 31 STEP 1 : NN=INT(N)
200 V=V-D*V-EFQ*X-A*SIN(WA*N): X=X+V
210 : IF ABS(X)>XM*100 THEN 250: REM * FEHL
      ER VERMEIDEN *
220 IF C=0 THEN DRAW 1,NN,100-X/XM: C=1: GOT
      O 240
230 DRAW 1,NN-1,100-XL/XM,NN,100-X/XM
240 : XL=X: REM * LETZTES X MERKEN *
250 NEXT
260 : GOTO 260: REM * WARTESCHLEIFE. AUSSTI
      EG MIT STOP-TASTE *
270 :
280 :
290 :
```

Listing 1b. Simulation für C 16 und C 128

System (Bild 7), damit seine Vibrationen sich nicht auf den Fußboden übertragen. Die Vibrationsdämpfung ist um so besser, je tiefer die Eigenfrequenz des Systems unter der Betriebsfrequenz des Motors liegt. Die Eigenfrequenz muß also beim Anlassen des Motors in jedem Fall durchlaufen werden.

```
130 EFQ=.01:D=0:A=.5:WA
      =(siehe Text)
140 V=0 und X=0.
150 XM=10:C=0
200 V=V-D*V-EFQ*X-A*SIN(WA*N)
      :X=X+V
```

Denken Sie bitte daran, in Zeile 200 das Sinusglied wiederherzustellen.

Um das Anfahren zu simulieren, geben Sie der Anre-




Bild 6. Einschwingen einer Waage




Bild 8. Erzwungene Schwingung

### So läuft eine Computer-Simulation ab:

Man sucht sich ein mathematisches Modell, welches das zu untersuchende Problem beschreibt (hier die Schwingungsgleichung). Diese Mathematische Gleichung bringt man in eine computergerechte Form (Zeile 200) und spielt eine Reihe von Möglichkeiten durch, bis man die günstigste Lösung durch Probieren findet.

Die in unserem Beispiel verwendete Schwingungsgleichung lautet in der gewohnten Form:

$$\frac{dV}{dt} + DV + \omega_E^2 \cdot X = A \cdot \sin \omega_A \cdot t$$

Der Bewegungszustand zur Zeit  $t=0$  ist durch die Anfangsbedingungen  $V_0$  und  $X_0$  gegeben. Unter der Voraussetzung, daß sich  $V$  und  $X$  in der sehr kurzen Zeit  $dt$  nicht wesentlich ändern, gilt in erster Näherung für  $t=dt$

$$V_1 = V_0 + \frac{dV}{dt} \cdot dt$$

und

$$X_1 = X_0 + \frac{dX}{dt} \cdot dt = X_0 + V_1 \cdot dt.$$

Da  $dV/dt$  durch die Differentialgleichung gegeben ist, kann man  $V_1$  und damit  $X_1$  berechnen. In gleicher Weise errechnet man  $V_2$  und  $X_2$  zur Zeit  $t = 2 \cdot dt$  aus  $V_1$  und  $X_1$ . Allgemein gilt für die Zeit  $t = N \cdot dt$

$$\frac{dV}{dt} = \frac{V_{N+1} - V_N}{dt} = -D \cdot V_N - \omega_E^2 \cdot X_N - A \cdot \sin \omega_A \cdot N \cdot dt.$$

oder

$$V_{N+1} = V_N - (D \cdot V_N + \omega_E^2 \cdot X_N + A \cdot \sin \omega_A \cdot N \cdot dt) \cdot dt.$$

Das Zeitintervall  $dt$  muß so klein gemacht werden, daß sich in ihm die Funktion nur wenig ändert. Wir haben  $dt = 1$  gewählt. Bei der Eigenfrequenz 0.01 wird dann eine Schwingungsperiode durch 63 Punkte dargestellt. Beachtet man noch, daß man in der Basic-Schreibweise die Indizes weglassen kann, so ergibt sich:

$$V = V - D \cdot V - \omega_E^2 \cdot X - A \cdot \sin \omega_A \cdot N \text{ und } X = X + V.$$

Ein Vergleich mit Zeile 200 macht klar, warum wir  $\omega_0^2 = \text{EFQ}$  und  $\omega_A = \text{WA}$  genannt haben.

Die Eulersche Methode eignet sich nicht nur zur Lösung der Schwingungsgleichung, sondern zur Integration aller Differentialgleichungen mit Anfangsbedingungen, die sich nicht exakt lösen lassen.

gungsfrequenz nacheinander folgende Werte:

$\text{WA} = 0.5 \cdot \text{SQR}(\text{EFQ})$ ,  
 $\text{SQR}(\text{EFQ})$  und  $2 \cdot \text{SQR}(\text{EFQ})$ .

Bild 8 zeigt die zugehörigen Einschwingvorgänge. Bei  $\text{WA} = \text{SQR}(\text{EFQ})$  (Kurve B) wird das Schwingungssystem im Takt seiner Eigenfrequenz angeregt und schaukelt sich zu großen Schwingungsweiten auf. Man nennt diese Erscheinung Resonanz. Dabei können bei kleiner Dämpfung so große Kräfte auftreten, daß das System zerstört wird. Bei kleineren (Kurve A) und größeren (C) Frequenzen halten sich die Schwingungsweiten in Grenzen.

Was kann man tun, um eine Resonanzkatastrophe zu verhindern? Kurve B zeigt, daß die Schwingung zum Aufschaukeln eine gewisse Zeit braucht. Sorgt man dafür,

daß die Motordrehzahl rasch vergrößert wird, dann kommt man auch bei mäßiger Dämpfung heil über die Resonanzstelle hinweg.

Mit den drei Simulationsbeispielen aus dem Bereich der mechanischen Schwingungen haben wir unsere Zauberformel bei weitem nicht ausgeschöpft. Wenn wir den Koeffizienten in Zeile 130 eine andere physikalische Bedeutung geben, können wir zum Beispiel das Einschwingverhalten von elektrischen Filtern nachbilden.

Sie haben sicher schon erkannt, daß sich dieses Basic-Programm vorzüglich zur Veranschaulichung und Erklärung von Schwingungsvorgängen eignet und daher sehr gut im Unterricht eingesetzt werden kann.

(Thomas Beyer/  
 Dr. Marie-Luise Beyer/do)




Bild 7. Motor auf Federdämpfung





Bild 4. Stoßdämpferschwingung

```

100 : REM* SIMULATION VON SCHWINGUNGSVORGAENGEN *
105 : REM* FUER C64 BITTE VORHER SIMON'S BASIC LA
DEN *
110 : REM* ODER GRAPHIKBEFEHLE AN EIGENES SYSTEM
ANPASSEN *
113 :
115 :
120 : REM* KENNGROESSEN DES SCHWINGUNGSSYSTEMS *
130 EFQ=.01:D=0:A=0:WA=0
140 V=0:X=80: REM* ANFANGSBEDINGUNGEN *
150 : XM=1:C=0: REM* X-MASSSTAB *
160 : HIRES1,6: REM* GRAPHIK EINSCHALTEN *
170 : LINE0,100,319,100,1: REM* ABSZISSE *
180 : LINE0,0,0,199,1: REM* ORDINATE *
190 FOR N=0TO319
200 V=V-D*V-EFQ*X-A*SIN(WA*N):X=X+V
210 : IFABS(X)>XM*100THEN250: REM* ERROR VERMEIDE
N*
220 : IFC=0THENPLOTN,100-X/XM,1:C=1:GOTO240: REM*
1.PUNKT SETZEN *
230 : LINEN-1,100-XL/XM,N,100-X/XM,1: REM* PUNKTE
VERBINDEN *
240 : XL=X: REM* LETZTES X MERKEN *
250 NEXT
260 : GOTO260: REM* WARTESCHLEIFE. AUSSTIEG MIT S
TOP-TASTE *

READY.

```

Listing 1a. Simulation für C 64 mit Simons-Basic